Здравствуйте, подскажите пожалуйста, не пойму чего не хватает.

Есть отчет в Ут 11, Карточка расчетов с клиентами(Продажи - отчеты по продажам - карточка расчетов с клиентами).

Пытаюсь программно его сформировать и вывести в реквизит типа табличный документ.

вот код:

```1c
    ТД = Новый ТабличныйДокумент;

    НашОтчет = Отчеты.КарточкаРасчетовСКлиентами24.ПолучитьМакет("ОсновнаяСхемаКомпоновкиДанных");
    НастройкиОСКД = НашОтчет.ВариантыНастроек.Карточка.Настройки;

    ПараметрыДанныхОСКД = НастройкиОСКД.ПараметрыДанных.Элементы;
    ЭлементНачалоПериода = ПараметрыДанныхОСКД.Найти("Период");
    ЭлементНачалоПериода.Использование = Истина;
    ЭлементНачалоПериода.Значение.ДатаНачала= НачалоМесяца(ТекущаяДата());
    ЭлементНачалоПериода.Значение.ДатаОкончания= КонецМесяца(ТекущаяДата());
    //Помещаем в переменную данные о расшифровке данных
    ДанныеРасшифровки = Новый ДанныеРасшифровкиКомпоновкиДанных;

    //Формируем макет, с помощью компоновщика макета
    КомпоновщикМакета = Новый КомпоновщикМакетаКомпоновкиДанных;

    //Передаем в макет компоновки схему, настройки и данные расшифровки
    МакетКомпоновки = КомпоновщикМакета.Выполнить(НашОтчет, НастройкиОСКД, ДанныеРасшифровки);

    //Выполним компоновку с помощью процессора компоновки
    ПроцессорКомпоновкиДанных = Новый ПроцессорКомпоновкиДанных;
    ПроцессорКомпоновкиДанных.Инициализировать(МакетКомпоновки, , ДанныеРасшифровки, Истина);
    //ПроцессорКомпоновкиДанных.Инициализировать(МакетКомпоновки, , , Истина);

    //Очищаем поле табличного документа
    Результат = ТД;

    //Выводим результат в табличный документ
    ПроцессорВывода = Новый ПроцессорВыводаРезультатаКомпоновкиДанныхВТабличныйДокумент;
    ПроцессорВывода.УстановитьДокумент(Результат);

    ПроцессорВывода.Вывести(ПроцессорКомпоновкиДанных);
```

В итоге у меня пустой отчет.
Если сформирую обычный способом, указав этот же период, тобишь август. все ок.
Сформирую без отбора тоже все ок.
Убрав период в коде также выдает пустой результат.

Что делаю не так?
---
Проверь процедуру ПриКомпоновкеРезультата в исходном отчете, в ERP в этом отчете выполняется ручная компоновка результата с подменой текста запроса.

В этом случае при программном формировании отчета необходимо выполнить те же действия, подменить запрос и все остальные манипуляции
---
попробуй передать данные в пользовательские настройки
---
По коду в топике все ок. Возможно вы не возвращаете табличный документ на клиент?
---
Возвращаю, это функция, она возвращает Результат на клиент, а там я ее записываю в таб. документ.
В итоге у меня там все показывает(шапку, параметры, колонки), но данных нет.
---
вот в чем дело. Т.е. скд отработал в "пустую". Тогда остается копать в сами настройки СКД. Возможно не все параметры заданы или есть отбор на пустое значение/ссылку. (5) Ирик дело говорит. В "новейших" БСП сам отчет не хранил выборки данных, там в основном лежит "рисунок" отчета (как завещал нам дядя Джек), а в ПриКомпоновке в отчет вставляется сам итоговый текст запроса.
---
Проверь процедуру ПриКомпоновкеРезультата в исходном отчете, в ERP в этом отчете выполняется ручная компоновка результата с подменой текста запроса.

В этом случае при программном формировании отчета необходимо выполнить те же действия, подменить запрос и все остальные манипуляции
---
```1c
    ТД = Новый ТабличныйДокумент;

    СтандартнаяОбработка = Ложь;
    ПользовательскиеНастройкиМодифицированы = Ложь;

    ДанныеРасшифровки = Неопределено;
    КомпановщикНастроек = Неопределено;

    СоздОтчет = Отчеты.КарточкаРасчетовСКлиентами24.Создать();

    КомпоновщикНастроек = СоздОтчет.КомпоновщикНастроек;

    УстановитьОбязательныеНастройки(КомпоновщикНастроек, ПользовательскиеНастройкиМодифицированы);

    // Сформируем отчет
    НастройкиОтчета = КомпоновщикНастроек.ПолучитьНастройки();

    ПараметрДанныеОтчета = КомпоновкаДанныхКлиентСервер.ПолучитьПараметр(НастройкиОтчета,"ДанныеОтчета");

    СоздОтчет.СхемаКомпоновкиДанных.НаборыДанных.НаборДанных.Запрос = ТекстЗапроса(ПараметрДанныеОтчета.Значение);

    КомпоновщикМакета = Новый КомпоновщикМакетаКомпоновкиДанных;
    МакетКомпоновки = КомпоновщикМакета.Выполнить(СоздОтчет.СхемаКомпоновкиДанных, НастройкиОтчета, ДанныеРасшифровки);

    ПроцессорКомпоновки = Новый ПроцессорКомпоновкиДанных;
    ПроцессорКомпоновки.Инициализировать(МакетКомпоновки, , ДанныеРасшифровки, Истина);

    ПроцессорВывода = Новый ПроцессорВыводаРезультатаКомпоновкиДанныхВТабличныйДокумент;
    ПроцессорВывода.УстановитьДокумент(ТД);

    ПроцессорВывода.Вывести(ПроцессорКомпоновки);

    Результат = ТД;

    Возврат Результат;
```

.Да, все получилось, спасибо.)